
NOTE: This worksheet demonstrates the use of Maple to illustrate the convergence of the roots using 
the Newton-Raphson method of finding roots of a nonlinear equation.

Introduction
Newton-Raphson method [text notes][PPT] is based on the principle that if the initial guess of the 
root of f(x) = 0 is at ix , then if one draws the tangent to the curve at f( ix ), the point 1+ix  where the 
tangent crosses the x-axis is an improved estimate of the root. Using the definition of the slope of a 
function,

)('
)(

1
i

i
ii xf

xfxx −=+  

The following simulation illustrates the convergence of the roots using Newton-Raphson method of 
finding roots of a nonlinear equation.

> restart;

Section I : Data.
The following is the data that is used to solve the nonlinear equation which is obtained from the 
floating ball problem from the General Engineering to find the depth 'x' to which the ball is 
submerged under water 

Function in f(x)=0
> f(x):=x^3-0.165*x^2+3.993*10^(-4):
Initial guess
> x0:=0.05:
Upper bound of range of 'x' that is desired
> uxrange:=0.12:
Lower bound of range of 'x' that is desired
> lxrange:=-0.02:
Maximum number of iterations
> nmax:=5:
Enter the umber of the root desired
> rootnumber:=1:



Section II: Before you start.
The derivative of the given function is found before we start.

> g(x):=diff(f(x),x);

 := ( )g x  − 3 x2 0.330 x

We now plot the data. The following function determines the upper and lower ranges on the Y-axis. 
This is done using the upper and lower ranges of the X-axis specified, and the value of the original 
functional at these values.

> yranger:=proc(uxrange,lxrange) 
local i,maxi,mini,tot; 
maxi:=eval(f(x),x=lxrange); 
mini:=eval(f(x),x=lxrange); 
for i from lxrange by (uxrange-lxrange)/10 to uxrange do 
if eval(f(x),x=i)<mini then mini:=eval(f(x),x=i) end if; 
if eval(f(x),x=i)>maxi then maxi:=eval(f(x),x=i) end if; 
end do; 
tot:=maxi-mini; 
-0.1*tot+mini..0.1*tot+maxi; 
end proc:

> yrange:=yranger(uxrange,lxrange):
> xrange:=lxrange..uxrange:

The following calls are needed to use the plot function
> with(plots):
Warning, the name changecoords has been redefined 

> with(plottools):
Warning, the name arrow has been redefined 

> plot(f(x),x=xrange,y=yrange,title="Entered function on given 
interval",legend=["Function"],thickness=3);



Section III: True Value.
The "true" solution is taken as the solution that Maple's numerical root solver obtains.  This is a 
decent assumption because their subroutines have been professionally written.  You must take 
caution, however, because Maple's "RootOf" function might be finding another of the function's 
roots.  For the rest of the sheet to be correct, you need to ensure that xrtrue is the root that you are 
attempting to find.  This can be altered by changing the value of 'rootnumber' above. 

> xrtrue:=RootOf(f(x),x,index=rootnumber):
> xrtrue:=evalf(xrtrue);



 := xrtrue 0.06237758151

Section IV: Value of root as a function of iterations.
Here the Newton-Raphson method algorithm is applied to generate the values of the roots, true 
error, absolute relative true error, approximate error, absolute relative approximate error, and the 
number of significant digits at least correct in the estimated root as a function of number of 
iterations.
> xr:=proc(n) 

     local p, q, i; 
     p:=x0; 
     q:=x0; 
     for i from 1 to n do 
         p:=q-eval(f(x),x=q)/eval(g(x),x=q); 
         q:=p; 
     end do; 
                
     p;      
end proc:

> nrange:=1..nmax:

Absolute true error
> Et:=proc(n) 

   abs(xrtrue-xr(n)); 
end proc:

Absolute relative true error
> et:=proc(n) 

   abs(Et(n)/xrtrue)*100; 
end proc:

Absolute approximate error
> Ea:=proc(n) 

   abs(xr(n)-xr(n-1)); 
end proc:

Absolute relative approximate error
> ea:=proc(n) 

   local p;    



   if n <=1 then 
      p:=0; 
   else 
      p:=abs(Ea(n)/xr(n))*100; 
   end if; 
p; 
end proc:   

Significant digits at least correct
> sigdigits:=proc(n) 

   local p; 
   p:=floor((2-log10(ea(n)/0.5))); 
   if p<0 then p:=0 end if; 
   p; 
end proc:

Section V: Graphs of Results.
> plot(xr,nrange,title="Estimated root as a function of number of 

iterations",thickness=3,color=red);



> plot(Et,nrange,title="Absolute true error as a function of 
number of iterations",thickness=3,color=navy);



> plot(et,nrange,title="Absolute relative true error as a 
function of number of iterations",thickness=3,color=blue);



> plot(Ea,nrange,title="Absolute approximate error as a function 
of number of iterations",thickness=3,color=aquamarine);



> plot(ea,nrange,title="Absolute relative approximate error as a 
function of number of iterations",thickness=3,color=green);



> plot(sigdigits,nrange,title="Number of significant digits at 
least correct as a function of number of 
iterations",thickness=3,color=gold);



> 

Section VI: Conclusion.

Maple helped us to apply our knowledge of numerical methods of finding roots of a nonlinear 
equation using the Newton-Raphson method to simulate the convergence of the root of the given 
nonlinear equation.

